
ABSTRACT 
In recent years, latent growth curve (LGC) modeling has become one of the most promising statistical techniques for 

modeling longitudinal data. The CALIS procedure in SAS® 9.3 could be used to fit an LGC model. As one application 

of structural equation modeling (SEM), LGC modeling relies on indices to evaluate model fit. However, it has been 

pointed out that when you are obtaining incremental fit indices, the default baseline model used in many popular SEM 

software packages, including PROC CALIS, is generally not appropriate for LGC models (Widaman and Thompson 

2003). This paper illustrates the basic framework of an LGC model and introduces a SAS macro, %LGCM, that fits a 

latent growth model and computes incremental fit indices based on more appropriate baseline models. 

 

INTRODUCTION 
Latent growth curve (LGC) modeling has become a popular statistical technique for modeling longitudinal data. Its 

capability and flexibility has earned it a great amount of attention in many fields, such as educational assessment, 

social science, clinical research, just to name a few. Perhaps the greatest advantage of LGC modeling is to model 

growth trajectory over time at both the group and individual level, which is not feasible with traditional repeated 

measures analysis.  

 

Since SAS/STAT 9.22, SAS has combined both the CALIS and TACLIS procedure. As its predecessors, the new 

CALIS procedure in SAS 9.3 is a general procedure for structural equation modeling (SEM). LGC models, being a 

special case of SEM, can be fitted in SAS by using PROC CALIS. The framework of a LGC model, however, is 

somewhat different from a standard confirmatory factor analysis (CFA). Subsequently, several authors (Preacher 

2010; Widaman and Thompson  2003) have recommended that when computing incremental fit indices, the default 

baseline model used by most popular SEM software packages, including PROC CALIS, is not appropriate for LGC 

models and an alternative baseline model should be used instead. 

 

The purpose of the current paper is: (1) to illustrate the basic framework of LGC modeling; (2) to explain the 

difference between the default baseline model used in PROC CALIS and the appropriate baseline model suggested 

by the literature; and (3) to offer a versatile SAS macro for practitioners with basic SAS skills to fit LGC models and 

obtain incremental fit indices that have been computed relative to an appropriate baseline model.  

 

LATENT GROWTH CURVE MODELING: AN ILLUSTRATED EXAMPLE 

For readers who are familiar with CFA, LGC models could be considered as a special case of it. To facilitate the 

introduction of LGC modeling, the dental measurement data from the famous Potthoff-Roy article (1964) are used. 

The data consist of dental measurements from 16 boys and 11 girls. The distances (mm) from the center of the 

pituitary gland to the pteryomaxillary fissure were measured at ages 8, 10, 12, and 14. 

 

UNCONDITIONAL MODEL 

First, a linear, unconditional growth model is introduced. At individual level, the model could be expressed in the 

following equation: 

Thus, a child’s measurement could be treated as a linear function of the two growth parameters (i.e., INTERCEPT 

and SLOPE) and random errors (E1-E4). INTERCEPT represents an individual’s starting point whilst SLOPE 

represents the growth trajectory. In LGC modeling, the two growth parameters are treated as latent variables (factors) 

and the repeated measurements are represented as indicators of the two factors. It is worth noting that the loadings 

of the time matrix are specified a priori in LGC modeling to reflect the hypothesized “shape” of change. That 

distinguishes LGC modeling from standard CFA/SEM models where usually loadings are estimated instead of being 

fixed. In the current example, the loadings of Slope are fixed as 0, 1, 2, 3, implying the shape of the growth 

trajectories is linear. Though the values of the factor loadings are arbitrary, the interpretation of the result might 

change according to the values the researchers choose (Hancock and Lawrence 2006; Stull 2008). Figure 1 

illustrates the path diagram of a linear LGC model for the current example. 

 

 
Figure 1. Path Diagram for the Dental Measurement Data Example 

 

Note that both the mean and covariance structure are analyzed simultaneously in LGC modeling.  In standard 

CFA/SEM models, only the covariance structure is usually examined (though there are some occasions when latent 

means might be of interest in CFA or SEM studies.). In LGC modeling, the latent mean of INTERCEPT represents 

the average initial status for the whole sample, and the latent mean of SLOPE describes the average growth across 

time periods at group level.   

The following shows the example SAS code for fitting an unconditional LGC model: 

 

title "Potthoff & Roy Data - Dental Measurements"; 

data dental; 

 input gender y1 y2 y3 y4; 

 cards; 

0 26.0 25.0 29.0 31.0 

0 21.5 22.5 23.0 26.5 

0 23.0 22.5 24.0 27.5 

0 25.5 27.5 26.5 27.0 

0 20.0 23.5 22.5 26.0 

0 24.5 25.5 27.0 28.5 

0 22.0 22.0 24.5 26.5 

0 24.0 21.5 24.5 25.5 

0 23.0 20.5 31.0 26.0 

0 27.5 28.0 31.0 31.5 

0 23.0 23.0 23.5 25.0  

0 21.5 23.5 24.0 28.0 

0 17.0 24.5 26.0 29.5  

0 22.5 25.5 25.5 26.0  

Statistics and Data AnalysisSAS Global Forum 2013

 
 



3 

 

0 23.0 24.5 26.0 30.0 

0 22.0 21.5 23.5 25.0 

1 21.0 20.0 21.5 23.0 

1 21.0 21.5 24.0 25.5 

1 20.5 24.0 24.5 26.0 

1 23.5 24.5 25.0 26.5 

1 21.5 23.0 22.5 23.5 

1 20.0 21.0 21.0 22.5 

1 21.5 22.5 23.0 25.0 

1 23.0 23.0 23.5 24.0 

1 20.0 21.0 22.0 21.5 

1 16.5 19.0 19.0 19.5 

1 24.5 25.0 28.0 28.0 

; 

run; 

 

title "Potthoff & Roy Data - Unconditional Linear LGC Model"; 

proc calis data=dental method=ml;    

   lineqs 

      y1 = 0. * Intercept + f_alpha                + e1, 

      y2 = 0. * Intercept + f_alpha  +  1 * f_beta + e2, 

      y3 = 0. * Intercept + f_alpha  +  2 * f_beta + e3, 

      y4 = 0. * Intercept + f_alpha  +  3 * f_beta + e4; 

   variance 

      f_alpha f_beta, 

      e1-e4; 

   mean 

      f_alpha f_beta; 

   cov 

      f_alpha f_beta; 

run; 

 

LINEQS statement is used to fit a LGC model in PROC CALIS.. In standard latent means structure, “Intercept” is a 

default constant regressed on the endogenous variable (i.e., Y1-Y4) and is always estimated as a fixed parameter. In 

LGC modeling, however, the intercept is supposed to be random, rather than fixed. To achieve this goal, the default 

“Intercept” has to be set to zero and a random intercept “f_alpha” is introduced. The VARIANCE, MEAN, and COV 

statements are used to specify which parameters we would like to estimate. Table 1 shows the maximum likelihood 

(ML) parameter estimates of the growth data set. 

 

Parameter Unstandardized SE Standardized 

Mean structure 

Latent means    

   INTERCEPT (f_alpha) 21.989
a
 0.418  

   SLOPE (f_beta) 1.362 0.145  

Covariance structure 

Latent factors    
   INTERCEPT 3.270 1.319  
   SLOPE 0.348 0.219  
   INTERCEPT , SLOPE 0.148 0.360 0.139 
Error variance    
   E1 2.185 0.988  
   E2 1.517 0.559  
   E3 2.403 0.809  
   E4 0.318 0.861  

a. Parameters that are significant at .05 printed in boldface. 

Table 1. ML Parameter Estimates for an Unconditional LGC Model of the Potthoff-Roy Data 

 

The estimated mean of INTERCEPT (f_alpha) is 21.989, which is a bit different from the observed mean of Y1 

(22.185). The estimated mean of SLOPE (f_beta) is 1.362, indicating the average increase of a child’s  pituitary-

pteryomaxillary distance over one year. 

The estimated variance of INTERCEPT is 3.270, suggesting the children are not homogeneous on this trait. The 

correlation between INTERCEPT and SLOPE is 0.139, indicating that children’s initial measurements are positively 

correlated with growth rates. That is, a child with longer pituitary-pteryomaxillary distance to begin with tends to grow 

faster. Figure 2 presents the ordinary least squares (OLS) regression lines of 27 children in the data set. 

 

 
Figure 2. OLS Trajectories for the 27 Children 

 

CONDITIONAL MODEL 

It is not uncommon to incorporate covariates to explain variance in INTERCEPT and SLOPE. In the current example, 

gender could be included into the model as a time-invariant covariate. The following shows the example code for 

fitting a conditional LGC model: 

 
title "Potthoff & Roy Data - Conditional Linear LGC Model with gender as the time-

invarying covariate"; 

proc calis data=dental method=ml; 

 lineqs 

  y1 = 0. * Intercept + f_alpha                + e1, 

        y2 = 0. * Intercept + f_alpha  +  1 * f_beta + e2, 

        y3 = 0. * Intercept + f_alpha  +  2 * f_beta + e3, 

        y4 = 0. * Intercept + f_alpha  +  3 * f_beta + e4, 

  f_alpha = b1*Intercept + c1*gender + e5, 

  f_beta  = b2*Intercept + c2*gender + e6; 

 variance 

  gender, 

  e1-e6; 

 mean 

  gender; 

 cov 

     e5 e6; 

run; quit; 

 

The ML parameter estimates of this conditional model are presented in Table 2. The mean of the covariate, GENDER 

(0.2507), is the same as the observed mean. The latent factor means of INTERCEPT (22.5049) and SLOPE (1.6513) 

are estimates for boys, since they are coded as 0 in this example.  

 

For the covariance structure, the correlation between INTERCEPT and SLOPE is -0.0891, indicating that initial status 

is negatively correlated with rate of growth, after incorporating GENDER into the model. The random intercept and 

slope (-1.2853 and -0.6942, respectively) are the effects of being female on those growth parameters. Thus, the 

estimated pituitary-pteryomaxillary distances for a 10-year-old girl is 21.2196 (=22.5049-1.2853), and the estimated 

yearly growth is 0.9571(=1.6513-0.6942). Since the values are negative, it means that both the intercept and slope 

are lower for girls than boys. It is clear from Figure 3 that girls tend to have shorter distances in the beginning and 

their rates of growth are slower compared to boys. 

 

Parameter Unstandardized SE Standardized 

Mean Structure    

Covariate mean    

GENDER 0.4074 0.098  

Latent factor means    

INTERCEPT (f_alpha) 22.5049 0.513  

SLOPE (f_beta) 1.6513 0.166  

Covariance structure    

Covariate variance    

GENDER 0.2507 0.070  

Disturbance variance & covariance    

INTERCEPT 2.8467 0.324  

SLOPE 0.272 0.188  

INTERCEPT, SLOPE -0.101 0.324 -0.0891 

Random intercept and slope    

GENDER on INTERCEPT -1.2853 0.795  

    GENDER on SLOPE -0.6942 0.257  

Error variance    

E1 2.228 0.996  

E2 1.485 0.545  

E3 2.503 0.805  

E4 0.098 0.738  

Table 2. ML Parameter Estimates for a Conditional LGC Model of the Potthoff-Roy Data 

 

 
Figure 3. The OLS Trajectories by Gender for the 27 Children  

 

INCREMENTAL FIT INDICES IN EVALUATING SEM MODEL FIT 
As with SEM in general, evaluating model fit is an important component in LGC modeling. A number of numeric fit 

indices have been proposed. The comparative fit index (CFI) and Tucker-Lewis index (TLI; also referred to as the 

nonnormed fit index) are two of the commonly-used fit indices in applied LGC research. Both have shown desirable 

properties in the simulation studies. In particular, they are not sensitive to sample size, which is a well-known 

weakness of the    statistic (Sun 2005).   

 

Both CFI and TLI belong to the family of incremental fit indices, which evaluate the relative improvement of the fit 

when comparing the fit of the hypothesized model to that of a more restrictive baseline model. The CFI and TLI can 

be calculated as follows: 

 

      
       

         

       
          

         
 (1) 

  

    
  

        
     

  
       

 (2) 

 

where   
  and     are the   statistic and degrees of freedom, respectively, for the baseline model while   

  and     are 

the counterparts for the hypothesized model. 

 

In most of the SEM software packages, such as LISREL, Mplus, and PROC CALIS, the baseline model used for 

calculating the incremental fit indices is the “independence null model”. In the independence null model, the 

covariances of the indicators are fixed to zero while only the variances are estimated. Usually this independence null 

model is fitted by setting all the paths from the factors to the corresponding indicators to 1, setting the covariance 

matrix of the random errors as a zero matrix, fixing all off-diagonal elements of the covariance matrix among the 

factors to be zero, and allowing the diagonal elements of the same covariance matrix to be free parameters (Rigdon 

1998a). 

 

The rationale of using the independence null model as the baseline model is to examine the relative improvement by 

comparing the hypothesized model with the worst possible model (Miles and Shevlin 2007). Nevertheless, it has been 

argued that the independence null model is impractical and might not always be appropriate for calculating 

incremental fit indices (Kline 2010; Marsh 1998; Ridgon 1998a; Ridgon 1998b; Sobel and Bohrnstedt 1985).  

 

APPROPRIATE BASELINE MODEL IN LGC MODELING  

To explain the appropriate baseline model for LGC modeling, assume we are fitting a three-time-point, unconditional 

linear LGC model (see Figure 4). When obtaining the incremental fit indices, SAS is actually fitting the baseline model 

as if the model was a two-factor CFA model rather than a LGC model (see Figure 5). The two growth parameters, 

INTERCEPT and SLOPE, are treated as factors here and all the path loadings are fixed to 1, and all the elements in 

the covariance matrix among the factors are fixed to zero.  

 

There are several problems with this default baseline model for LGC modeling: (1) the loadings from SLOPE are fixed 

to one, which is not reasonable in the LGC modeling context; (2) the latent mean structure is entirely ignored in this 

case; and consequently, (3) if the mean structure is not accounted for, it will impact on the calculation of the number 

of the parameters. When the mean structure is not analyzed (such as the standard CFA models), the number of 

parameters equals to p(p+1)/2, where p is the number of indicators. However, the number of parameters is to 

p(p+3)/2 when the mean structure is incorporated. Thus, for a three-time-point, unconditional linear LGC model, the 

number of parameters should be 9, but the default baseline model fit by PROC CALIS is 6.  

 

Therefore, the default independence null model is obviously not appropriate for LGC modeling. It has been suggested 

that a more appropriate baseline model for LGC modeling should be an intercept-only model in which no growth is 

allowed over time (Preacher, Wichman, MacCallum, and Briggs 2008; Widaman and Thompson 2003). This 

intercept-only model is specified such that (1) the mean of the intercept is freely estimated; (2) the variance of the 

intercept is fixed to zero; and (3) all the error variances of the indicators are constrained to be equal and freely 

estimated (see Figure 6). Since the intercept-only model is not built into most of the popular SEM programs, the 

incremental fit indices must be calculated by hand (Preacher 2010).    

Figure 4. A Three-time-Point, Linear LGC Model          Figure 5. Independence Null Model 

 

 

 
Figure 6. Intercept-only Model 

 

MACRO %LGCM  

As discussed earlier, the default baseline model used in SAS is not appropriate for LGC models. Thus, a SAS macro 

%LGCM (the syntax is shown in the Appendix) was developed to compute the incremental fit indices by using the 

intercept-only model described above.  

 

FEATURES OF %LGCM: 

1. Ease of use in its simplest form; 

2. Allows polynomial trajectories: quadratic, cubic, etc; 

3. Allows user-specified loadings for the “time matrix”; 

4. Allows constraints on the equality of the error variances. 

 

REQUIREMENTS: 

1. %LGCM can only be used in SAS 9.3, which is due to the new features and syntax in PROC CALIS. 

2. The indicators used in %LGCM must follow certain naming convention. All the indicators should have some 

prefix, followed by numeric suffix 1, 2, 3, etc. The number must start at 1 and end with the number of waves. 

Some examples are {y1, y2, y3} or {test1, test2, …, test5}. 

 

ADDITIONAL OPTION FOR CUSTOMIZED LOADINGS 

If one would like to specify the loadings in the time matrix, the loadings need to be separated by forward slashes (“/”). 

Further, plus and minus signs are needed before each value of the loadings. If the value of the loading is zero, then a 

plus sign is needed. Examples could be {-3/-2/-1/+0} or {-1.5/-.5/+0.5/+1.5}. 

 

SAS USER INPUT 

%macro lgcm(data=&syslast, method=ml, prefix=, wave=, polypower=, loading=, error=, 

print=); 

 

To invoke macro %LGCM, the arguments are described as follows, 

Parameter Description Default 

DATA= Name of the data set &syslast 

METHOD= Estimation method ML 

PREFIX= The prefix of the indicators (required)  

WAVE= Number of waves (required)  

POLYPOWER= Degree of polynomial. Number used to indicate the shape the growth 

trajectory.  

E.g., 1=Linear, 2=Quadratic 

1 

LOADING= User-specified loadings for the “time-matrix” (optional)  

ERROR= Restrict error variances to be equal or not (yes/no) NO 

PRINT= Display the output from CALIS procedure or not (yes/no) NO 

 

SAS MACRO OUTPUT 

                          Adjusted Incremental Fit Indices 

 

 

           Bentler Comparative Fit Index (CFI)                             0.9869 

           Bentler-Bonnet Normed Fit Index (NFI)                           0.9396 

           Bentler-Bonnet Non-normed Index (NNFI)                          0.9686 

           Bollen Normed Index Rho1 (aka Relative Fit Index; RFI)          0.8551 

           Bollen Non-normed Index Delta2 (aka Incremental Fit Index; IFI) 1.0643 

 

 

         The baseline model that was used to calculate the above values of the 

         incremental fit indices was an intercept-only model, which is considered 

         to be appropriate in the latent growth curve modeling context (see 

         Widaman & Thompson, 2003). 

Output 1. Output from %LGCM 

 

The SAS output above was obtained from fitting an unconditional model to the Potthoff-Roy data. %LGCM provides 

the following incremental fit indices computed from the intercept-only baseline model: 
1. The comparative fit index (CFI); 

2. The normed fit index (NFI); 

3. The non-normed fit index (NNFI); 

4. Bollen’s normed index Rho1(aka Relative Fit Index; RFI); 

5. Bollen’s non-normed index Delta2 (aka Incremental Fit Index; IFI).  

 

It is worth noting that the adjusted values obtained from %LGCM are not necessarily lower or higher than the values 

obtained from the default baseline model. For example, the NNFI (.9778) obtained from the default baseline model is 

larger than the NNFI (.9686) obtained from %LGCM; on the other hand, the CFI (.9815) obtained from the default 

baseline model is smaller than the CFI (.9869) obtained from %LGCM.  

 

LIMITATIONS: 

1. The usage of %LGCM is currently limited to an unconditional growth model.  

2. The baseline model used in %LGCM is limited to the one described in Figure  6.  

 

For more complicated growth models, such as conditional growth models or multivariate growth models, users might 

specify the baseline model and compute the fit indices manually using equations (1) & (2). For a more detailed 

description, see Widaman and Thompson (2003).  

 

Furthermore, there is not a consensus in the literature as to what model is the most appropriate baseline for 

incremental fit indices in LGC modeling. The interested reader should consult Widaman and Thompson (2003). 

However, the intercept-only model used in %LGCM is conservative enough that should be suitable in most cases.  

   

CONCLUSION 
A SAS macro %LGCM for computing improved incremental fit indices for LGC modeling is presented in this paper. 

%LGCM calculates incremental fit indices by using the more appropriate baseline model than the default baseline 

model used in PROC CALIS. These modifying fit indices facilitate better evaluation of the fit of a LGC model to 

longitudinal data. 

APPENDIX 

MACRO %LGCM CODE 
********************************************************************* 

** data = Data set                                       * 

** method = Estimation method (default method: Maximum likelihood)  * 

** prefix = Prefix of the indicators                * 

** wave = Waves of the data       * 

** polypower = Degree of polynomial (e.g., 1=linear, 2=quadratic)   * 

** loading = User-defined loadings.            * 

**  - Note 1: Use "forward slash"(/) to separate loadings    * 

** - Note 2: Use "plus/minus" sign before each number           * 

** - Example: loading = -3/-2/-1/+0                             * 

** error = Restrict error variances to be equal or not (yes/no)     * 

** print = Display the output for CALIS procedure or not (yes/no)   * 

********************************************************************; 

 

options nodate nonumber ls=84 ps=64; 

*** Macro for hypothesized model; 

%macro Hyp(data=&syslast, method=ml, prefix=, wave=, polypower=1,  

           loading=, error=no, print=yes); 

 

 *** Suppress the output; 

 %if %upcase(&print)=NO %then %let noprint=noprint;  

 %else %let noprint=; 

 

 *** Define the parameters to be estimated for polynomial;  

 *** (for mean, variance, & cov statements); 

 %if &polypower %then %do; 

        %let params = f_alpha; 

        %do pow = 1 %to &polypower; 

             %let params = &params f_beta&pow ; 

        %end; 

 %end; 

 

 *** IndexCode: 203 Chi-square/204 Chi-square DF; 

 proc calis data=&data method=&method nostand noparmname  

       outfit=fitHyp(where=(IndexCode in (203,204))  

       keep=IndexCode FitIndex FitValue) &noprint ; 

 

  lineqs 

   /*** Loop through the equation for each time period. ***/ 

        %do i = 1 %to &wave; 

             &prefix&i = 0*Intercept + f_alpha 

    /*** User-defined loadings ***/ 

    %if %bquote(&loading) ^= %then %do;    

    %let coef = %qscan(&loading,&i,/); 

      &coef*f_beta1 

    %end; 

     /*** Standard loadings ***/ 

    %else %do;  

    %do j=1 %to &polypower; 

        + %eval((&i-1)**&j)*f_beta&j 

    %end; 

    %end; + e&i                          

   /*** Use commas to separate the equations ***/  

   %if &i < &wave %then %do; 

            , 

         %end; 

  /*** Put semi-colon at the end of the last equation ***/ 

      %else %do; 

      ; 

         %end; 

        %end; 

 

   variance 

    &params, 

    /*** Loop through the errors ***/  

    %do j = 1 %to &wave;  

     e&j  

       %end; 

    %if %upcase(&error)=YES %then %do; 

     = &wave*evar 

    %end; 

     ; 

   mean 

    &params; 

   cov 

    &params; 

   run; 

 

%mend; 

 

*** Macro for baseline model; 

%macro Base(data=&syslast, method=ml, prefix=, wave= ); 

 

 proc calis data=&data method=&method nostand noparmname  

       outfit=fitBase(where=(IndexCode in (203,204))  

       keep=IndexCode FitIndex FitValue) noprint; 

  lineqs 

 %do i = 1 %to &wave; 

  &prefix&i = 0*Intercept + f_alpha + e&i 

  /*** Use commas to separate the equations ***/  

   %if &i < &wave %then %do; 

            , 

         %end; 

  /*** Put semi-colon at the end of the last equation ***/ 

      %else %do; 

      ; 

         %end; 

 %end;  

 

  variance 

   f_alpha = 0, 

   e1-e&wave = &wave*evar; 

  mean 

   f_alpha; 

 run; 

 

%mend; 

 

%macro lgcm(data=&syslast, method=ml, prefix=, wave=, polypower=, loading=, error=, 

print=); 

 

 *** Check SAS version; 

 %if %sysevalf(&sysver < 9.3) %then %do; 

  %put ERROR: The LGCM macro requires SAS version 9.3; 

 %goto exit; 

 %end; 

 

 *** Call in the macros and calculate the value; 

 %Hyp(data=&data, method=&method, prefix=&prefix, wave=&wave, 

polypower=&polypower, loading=&loading, error=&error, print=&print); 

 %Base(data=&data, method=&method, prefix=&prefix, wave=&wave); 

 

 *** Transpose and rename the result file; 

 proc transpose data=fitHyp(drop=IndexCode) 

 out=fit(rename=(col1=chisq1 col2=DF1) drop=_name_ _LABEL_);  

 run; 

 

 *** Transpose and rename the result file; 

 proc transpose data=fitBase(drop=IndexCode)  

 out=fit0(rename=(col1=chisq0 col2=DF0) drop=_name_ _label_);  

 run; 

 

 *** Display the result in the Output window; 

 ods listing; 

 

 *** Merge results and calculate CFI and TLI; 

 data fit; 

  merge fit fit0; 

  TLI = ( (chisq0/df0) - (chisq1/df1) ) / ((chisq0/df0) - 1); 

  RFI = ( (chisq0/df0) - (chisq1/df1) ) / (chisq0/df0); 

  NFI = ( chisq0 - chisq1) / chisq0; 

  IFI  = (chisq0 - chisq1) / (chisq0 - df0); 

  d1 = chisq1 - df1; 

  d0 = chisq0 - df0; 

  RNI = 1 - ( d1/d0); 

  CFI = 1 - ( max(d1,0)/ max(d1,d0,0) ); 

 

  file print; 

  title3 "Adjusted Incremental Fit Indices"; 

  put  // 

  @12 "Bentler Comparative Fit Index (CFI)" @76 CFI  comma6.4/ 

  @12 "Bentler-Bonnet Normed Fit Index (NFI)" @76 NFI  comma6.4/ 

  @12 "Bentler-Bonnet Non-normed Index (NNFI)" @76 TLI comma6.4/ 

  @12 "Bollen Normed Index Rho1 (aka Relative Fit Index; RFI)" @76 RFI 

comma6.4/ 

  @12 "Bollen Non-normed Index Delta2 (aka Incremental Fit Index; IFI)" @76 

IFI comma6.4 / 

   // 

  @10 "The baseline model that was used to calculate the above values of 

the"  / 

  @10 "incremental fit indices was an intercept-only model, which is 

considered"  / 

  @10 "to be appropriate in the latent growth curve modeling context (see" 

/  

  @10 "Widaman & Thompson, 2003)."; 

 run; 

%exit:   *** EXIT: Label indicator; 

%mend; 

